

DEPARTMENT OF THE INTERIOR  
U.S. GEOLOGICAL SURVEY

In Situ Geomechanics of Crystalline and Sedimentary Rocks

Part VI: An Update on Two BASIC Computer Programs for the Determination  
of In Situ Stresses Using the CSIRO Hollow Inclusion Stress  
Cell and the USBM Borehole Deformation Gage

By

Henri S. Swolfs<sup>1</sup> and Philip S. Powers<sup>1</sup>

Open-File Report 85-509

1985

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards and stratigraphic nomenclature. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS.

<sup>1</sup>U.S. Geological Survey  
Denver, Colorado

## IN SITU GEOMECHANICS OF CRYSTALLINE AND SEDIMENTARY ROCKS

### PART VI: AN UPDATE ON TWO BASIC COMPUTER PROGRAMS FOR THE DETERMINATION OF IN SITU STRESSES USING THE CSIRO HOLLOW INCLUSION STRESS CELL AND THE USBM BOREHOLE DEFORMATION GAGE

By Henri S. Swolfs and Philip S. Powers

#### PREFACE

This report is the sixth of a series summarizing the results of the U.S. Geological Survey's research program in geomechanics aimed at investigating and assessing the potential of crystalline and sedimentary rock masses as geological repositories of nuclear waste. The first five parts of this series of reports are referenced below:

- Savage, W. Z., and Swolfs, H. S., 1980, the long-term deformation and time-temperature correspondence of viscoelastic rock--an alternative theoretical approach, Pt. 1 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 80-708, 21 p.
- Smith, W. K., 1982, Two BASIC computer programs for the determination of in situ stresses using the CSIRO hollow inclusion stress cell and the USBM borehole deformation gage [Pt. 2 of In situ geomechanics of crystalline and sedimentary rocks]: U.S. Geological Survey Open-File Report 82-489, 40 p.
- Swolfs, H. S., 1982, First experiences with the C.S.I.R.O. hollow-inclusion stress cell, Pt. 3 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File report 82-990, 20 p.
- Nichols, T. C., 1983, Continued field testing of the modified U.S. Geological Survey 3-D borehole stress probe, Pt. 4 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 83-750, 11 p.
- Savage, W. Z., Powers, P. S., and Swolfs, H. S., 1984, RVT--A FORTRAN program for an exact elastic solution for tectonic and gravity stresses of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 84-827, 12 p.

Published journal articles that report on the findings of this program are referenced below:

- Swolfs, H. S., and Kibler, J. D., 1982, A note on the Goodman Jack: Rock Mechanics, v. 15, no. 2, p. 57-66.
- Swolfs, H. S., 1983, Aspects of the size-strength relationship of unjointed rocks: Chapter 51 in Rock Mechanics--Theory-Experiment-Practice: 24th U.S. Symposium on Rock Mechanics, College Station, Texas, p. 501-510.
- Swolfs, H. S., 1984, The triangular stress diagram--a graphical representation of crustal stress measurements: U. S. Geological Survey Professional Paper 1291, 19 p.
- Swolfs, H. S., and Savage, W. Z., 1984, Site characterization studies of a volcanic cap rock, Chapter 39 in Rock Mechanics in Productivity and Protection: 25th U.S. Symposium on Rock Mechanics, Evanston, Illinois, p. 370-380.

## CONTENTS

	Page
Introduction .....	1
Continuing documentation and program listings.....	2
References .....	19

Part VI: An Update on Two BASIC Computer Programs for the Determination  
of In Situ Stresses Using the CSIRO Hollow Inclusion Stress  
Cell and the USBM Borehole Deformation Gage

**INTRODUCTION**

The determination of the state of stress in rock masses is one of the more important problems in geomechanics, both for the design of underground openings and for assessing the effects of external influences, such as heating by stored radioactive materials. A common technique of stress determination is the overcoring method: a small-diameter pilot hole is drilled in the rock mass; an instrument for measuring strain or deformation in several directions is placed in the pilot hole; and a larger coaxial annulus is then drilled around the instrumented hole to register the deformations. Because the instrumented specimen is now released from the ambient stress field, the resulting deformations and knowledge of the rock moduli give a measure of the amount of stress relieved.

Two instruments in common use today are the U.S. Bureau of Mines (USBM) borehole deformation gage (Merrill, 1967), and the Commonwealth Scientific and Industrial Research Organization (CSIRO, Australia) hollow-inclusion stress cell (Worotnicki and Walton, 1976).

The USBM instrument measures the change in borehole diameter along three directions in a single plane perpendicular to the borehole axis. Measurements in three or more nonparallel boreholes are required for a complete three-dimensional stress determination. The CSIRO instrument uses three strain-gage rosettes mounted around the circumference of a hollow epoxy cylinder which is cemented to the borehole wall prior to overcoring. Because of the three-dimensional orientation of the nine strain gages, it is possible to obtain a complete stress determination in a single borehole. Rock-mass moduli can be measured with these instruments using specialized techniques (Fitzpatrick, 1962).

The mathematical solution for the stress components is complex (Panek, 1966) and a digital computer is required to facilitate the reduction of the field measurements. In the past, the usual procedure has been to manually record the data in the field, return to the office, and perform the data reduction using a FORTRAN program and a large mainframe computer. Recent advances in computer technology and the availability of desk-top microcomputers have made it possible to perform immediate data reduction onsite and, in addition, to record the instrument response in real time during the overcoring process. In the field, this capability allows the investigator to immediately analyze the data, to assess its quality, and to make decisions to either abandon a particular test or to conduct additional tests while the drilling crew is still onsite. Most microcomputers in common use are programmable in BASIC. This report updates two BASIC data-reduction programs, one for use with the USBM borehole deformation gage, and the other for the CSIRO hollow-inclusion stress cell. These specialized programs were originally developed by Smith (1982) and offer the following advantages over the previously available FORTRAN programs:

1. They can be run on a desk-top microcomputer which can be transported to the field site, allowing data analyses to be performed on site.
2. The programs are shorter, but more specialized, because the available built-in matrix routines in BASIC eliminate the need for many subroutines.
3. Data input is greatly simplified by assigning the known (standard) orientation angles to the individual strain gages in the CSIRO instrument.
4. The programs work with any consistent set of units (English or SI).

#### CONTINUING DOCUMENTATION AND PROGRAM LISTINGS

Programs continually evolve even after a standard version has been released; this evolution includes the development of new capabilities, the detection and repair of errors, and application-oriented modifications. Here, we report on updated versions of programs that were originally developed on the Hewlett-Packard (HP) 9800 series microcomputers (Smith, 1982). With some modifications, the BASIC programs have been adapted for use on the Hewlett-Packard series 200 technical computers which are powerful standalone workstations and instrument controllers.

The modifications are as follows:

1. The conversion of the programs USBM and CSIRO from the Hewlett-Packard 9845 BASIC to the Hewlett-Packard Series 200 BASIC 2.0 with an AP2.1 upgrade was straightforward and readily accomplished. The two major modifications were (a) to rename all simple variable names that were already in use as array variable names, and (b) to move the END statement so that it follows the final subroutine of the programs. The simple variable names that were renamed in both programs are K, J, and I; the variable K was replaced by K1, J by J1, and I by I1. The original END statement immediately preceding the final subroutine of the programs, was replaced by a REM statement and a GO TO statement that now direct the flow of the program to the new END statement immediately following the final subroutines.
2. Because BASIC 3.0 (or BASIC 2.0 plus AP2.1 upgrade) has built-in MAX and MIN functions, the 27 statements in the sort routine have been replaced by the 5 program lines suggested by Smith (1982, p. 18).

Definitions of program variables and program operation remain as described by Smith (1982). Memory requirements to run these upgraded programs on the HP 200 series computers are 24K bytes for USBM and 16K bytes for CSIRO.

```

10 REM ***
20 REM ***
30 REM *** *****
40 REM ***
50 REM *** A PROGRAM FOR THE CALCULATION OF STRESS COMPONENTS, PRINCIPAL
60 REM *** STRESSES, AND THEIR STANDARD DEVIATIONS FROM SETS OF READINGS
70 REM *** FROM THE u.s. bureau of mines borehole deformation gage IN
80 REM *** THREE OR MORE NON-PARALLEL BOREHOLES.
90 REM ***
100 REM *** PROGRAM MODIFIED FROM USBM BY WILLIAM K. SMITH BY P. POWERS
110 REM *** JUNE 7, 1985.
120 REM *** PROGRAM WILL RUN WITH THE HP-9836 WITH BASIC 3.0. OR BASIC
130 REM *** 2.0 WITH THE AP2_1 EXTENSION.
140 REM ***
150 REM *** LITTLE OR NO MODIFICATION WAS REQUIRED TO RUN THE PROGRAM ON
160 REM *** THE HP-9836. ONE SIGNIFICANT MODIFICATION WAS THAT THE HP-9836
170 REM *** DOES NOT DIFFERENTIATE BETWEEN ARRAY NAMES AND SIMPLE VARIABLE
180 REM *** NAMES THAT ARE THE SAME, THEREBY PRODUCING AN ERROR. NEW SIMPLE
190 REM *** VARIABLES HAVE BEEN INTRODUCED INTO THIS PROGRAM TO COMPENSATE
200 REM *** FOR THIS DIFFERENCE (i.e., VARIABLE 'I' WAS CHANGED TO 'I1').
210 REM ***
220 REM *** ***** REFERENCES *****
230 REM ***
240 REM *** PANEK, LOUIS A., 1966, Calculation of the average ground stress
250 REM *** components from measurements of the diametral deformation of
260 REM *** a drill hole: U.S. Bureau of Mines Rept. of Inv. 6732, 41 p.;
270 REM *** or ASTM S.T.P. 402. p. 106-132.
280 REM *** SMITH, WILLIAM K., 1982, Two BASIC computer programs for the
290 REM *** determination of in-situ stresses using the CSIRO hollow
300 REM *** inclusion gage and the USBM borehole deformation gage:
310 REM *** U.S. Geological Survey Open-File Rept. 82-489, 40 p.
320 REM ***
330 REM *** *****
340 REM ***
350 REM *** COORDINATE SYSTEM:
360 REM *** X-AXIS IS POSITIVE EAST
370 REM *** Y-AXIS IS POSITIVE NORTH
380 REM *** Z-AXIS IS POSITIVE UP
390 REM ***
400 REM *** H1-AXIS IS HORIZONTAL TO THE RIGHT WHEN LOOKING INTO THE HOLE
410 REM *** H2-AXIS COINCIDES WITH THE BOREHOLE AXIS
420 REM *** H3-AXIS IS IN A VERTICAL PLANE, POINTS GENERALLY UPWARD
430 REM ***
440 REM *** BEARINGS ARE IN DEGREES CLOCKWISE FROM NORTH (AZIMUTH).
450 REM *** INCLINATIONS ARE IN DEGREES FROM THE HORIZONTAL, POSITIVE UPWARD.
460 REM *** BEARING OF A VERTICAL HOLE IS 90 DEGREES COUNTER-CLOCKWISE FROM
470 REM *** THE AZIMUTH OF H1-AXIS, AS DETERMINED FROM THE ORIENTATION OF THE
480 REM *** SETTING TOOL.
490 REM ***
500 REM *** ***** UNITS *****
510 REM ***
520 REM ***
530 REM *** CALIB. FACTOR DIAMETER YOUNG'S MODULUS STRESS
540 REM *** -----
550 REM *** MICRO-IN/MICROSTRAIN INCHES 106PSI PSI
560 REM *** MICRO-M/MICROSTRAIN METERS GPa KPa
570 REM ***
580 REM *** *****
590 OPTION BASE 1
600 PRINTER IS 1

```

```

610 R1=57.29578
620 DIM A(6,6),B(6,1),C(6,6),F(50,4),G(6,1),H(3),I(3,3),J(50,6),K(50,9)
630 DIM O(50,7),P(3,3),S(3,3),T(3),U(50),V(6),X(3,3),Y(3,3),Z(3,3)
640 REM *** define measurement angles for each channel
650 T(1)=90
660 T(2)=150
670 T(3)=30
680 REM *** read calibration factors for each channel
690 PRINT "Enter calibration factors for channels 1-3 in"
700 PRINT "micro-length per microstrain"
710 INPUT H(1),H(2),H(3)
720 K9=1
730 M=2
740 M=M-1
750 PRINT CHR$(12)
760 PRINT "Enter bearing, inclination, and diameter of borehole."
770 PRINT "Young's modulus (in mega-stress), and poisson's ratio"
780 B1=0
790 I1=0
800 D1=0
810 E1=0
820 P1=0
830 INPUT B1,I1,D1,E1,P1
840 IF B1=999 THEN 1340
850 B1=B1/R1
860 I1=I1/R1
870 E=E1*1.0E+6
880 L1=COS(B1)
890 M1=-SIN(B1)
900 N1=0
910 L2=SIN(B1)*COS(I1)
920 M2=COS(B1)*COS(I1)
930 N2=SIN(I1)
940 L3=-SIN(B1)*SIN(I1)
950 M3=-COS(B1)*SIN(I1)
960 N3=COS(I1)
970 PRINT CHR$(12)
980 PRINT "Enter channel no. and microstrain reading"
990 PRINT "Enter '999.999' to terminate readings for this borehole"
1000 INPUT C1,U(M)
1010 M=M+1
1020 IF C1=999 THEN 760
1030 A1=T(C1)/R1
1040 O(M-1,1)=U(M-1)
1050 U(M-1)=U(M-1)*1.0E-6*H(C1)
1060 O(M-1,2)=C1
1070 O(M-1,3)=B1*R1
1080 O(M-1,4)=I1*R1
1090 O(M-1,5)=D1
1100 O(M-1,6)=E1
1110 O(M-1,7)=P1
1120 K(M-1,1)=L1
1130 K(M-1,2)=L2
1140 K(M-1,3)=L3
1150 K(M-1,4)=M1
1160 K(M-1,5)=M2
1170 K(M-1,6)=M3
1180 K(M-1,7)=N1
1190 K(M-1,8)=N2
1200 K(M-1,9)=N3

```

```

1210 F(M-1,1)=(D1*(1+2*COS(2*A1))*(1-P1*P1)+D1*P1*P1)/E
1220 F(M-1,2)=-D1*P1/E
1230 F(M-1,3)=(D1*(1-2*COS(2*A1))*(1-P1*P1)+D1*P1*P1)/E
1240 F(M-1,4)=D1*4*SIN(2*A1)*(1-P1*P1)/E
1250 J(M-1,1)=F(M-1,1)*L1*L1+F(M-1,2)*L2*L2+F(M-1,3)*L3*L3+F(M-1,4)*L1*L3
1260 J(M-1,2)=F(M-1,1)*M1*M1+F(M-1,2)*M2*M2+F(M-1,3)*M3*M3+F(M-1,4)*M1*M3
1270 J(M-1,3)=F(M-1,1)*N1*N1+F(M-1,2)*N2*N2+F(M-1,3)*N3*N3+F(M-1,4)*N1*N3
1280 J(M-1,4)=2*F(M-1,1)*L1*M1+2*F(M-1,2)*L2*M2+2*F(M-1,3)*L3*M3+F(M-1,4)*(L1*M
3+L3*M1)
1290 J(M-1,5)=2*F(M-1,1)*M1*N1+2*F(M-1,2)*M2*N2+2*F(M-1,3)*M3*N3+F(M-1,4)*(M1*N
3+M3*N1)
1300 J(M-1,6)=2*F(M-1,1)*N1*L1+2*F(M-1,2)*N2*L2+2*F(M-1,3)*N3*L3+F(M-1,4)*(N1*L
3+N3*L1)
1310 IF K9=2 THEN 1430
1320 PRINT CHR$(12)
1330 GOTO 980
1340 M=M-1
1350 PRINTER IS 701
1360 PRINT
1370 PRINT "U.S. BUREAU OF MINES BOREHOLE DEFORMATION GAGE DATA REDUCTION*"
1380 PRINT
1390 PRINT "number of deformation measurements = ";M
1400 PRINT
1410 PRINT " MICRO-          ***** BOREHOLE *****"
1420 PRINT " STRAIN    DEFORMATION CH  BEARING INCL.  DIAMETER  E      PR"
1430 FOR K1=1 TO M
1440 PRINT USING 1460:O(K1,1),U(K1),O(K1,2),O(K1,3),O(K1,4),O(K1,5),O(K1,6),O
(K1,7)
1450 NEXT K1
1460 IMAGE S6D,5X,S.6D,4X,D.3X,3D,D.3X,S2D,D.3X,D.4D,3X,2D,2D,3X,.3D
1470 PRINT
1480 PRINT "          Channel Angle  Calib. factor"
1490 FOR K1=1 TO 3
1500 PRINT USING 1520:K1,T(K1),H(K1)
1510 NEXT K1
1520 IMAGE 20X,D,5X,3D,D,5X,3D,3D
1530 PRINT
1540 REM *** CLEAR A(I,J) and G(J,I) ARRAYS
1550 FOR J1=1 TO 6
1560 G(J1,1)=0
1570 FOR I1=1 TO 6
1580 A(I1,J1)=0
1590 NEXT I1
1600 NEXT J1
1610 REM *** COMPUTE [A] AND [G] ARRAYS
1620 FOR K1=1 TO M
1630 FOR J1=1 TO 6
1640 G(J1,1)=G(J1,1)+U(K1)*J(K1,J1)
1650 FOR I1=1 TO 6
1660 A(I1,J1)=A(I1,J1)+J(K1,I1)*J(K1,J1)
1670 NEXT I1
1680 NEXT J1
1690 NEXT K1
1700 MAT C= INV(A)
1710 MAT B= C*G
1720 REM *** compute statistical parameters
1730 IF K9=2 THEN 1950
1740 S1=0
1750 FOR K1=1 TO M
1760 S1=S1+U(K1)*U(K1)

```



```

1770 NEXT K1
1780 S2=B(1,1)*G(1,1)+B(2,1)*G(2,1)+B(3,1)*G(3,1)+B(4,1)*G(4,1)+B(5,1)*G(5,1)+B
(6,1)*G(6,1)
1790 S3=S1-S2
1800 R2=1-S3/S1
1810 R=SQR(R2)
1820 S4=SQR(S3/(M-6))
1830 PRINT USING 1890:S1
1840 PRINT USING 1900:S2
1850 PRINT USING 1910:S3
1860 PRINT USING 1920:S4
1870 PRINT USING 1930:R2
1880 PRINT USING 1940:R
1890 IMAGE "DEFORMATION SUM OF SQUARES = ".D.4DE
1900 IMAGE " [B].[G] = ".D.4DE
1910 IMAGE " RESIDUAL SUM OF SQUARES = ".D.4DE
1920 IMAGE " STD. DEV. OF FITTED DATA = ".D.4DE
1930 IMAGE " GOODNESS OF FIT (R**2) = ".D.5D
1940 IMAGE " CORRELATION COEFFICIENT = ".D.5D
1950 FOR I1=1 TO 6
1960 V(I1)=S4*SQR(C(I1,I1))
1970 NEXT I1
1980 IF K9=2 THEN 2880
1990 PRINT
2000 PRINT "STRESS COMPONENTS          STD. DEV."
2010 PRINT USING 2020:B(2,1),V(2)
2020 IMAGE "      NORMAL N-S = ".S6D, 4X, 5D
2030 PRINT USING 2040:B(1,1),V(1)
2040 IMAGE "      NORMAL E-W = ".S6D, 4X, 5D
2050 PRINT USING 2060:B(3,1),V(3)
2060 IMAGE "      NORMAL VER = ".S6D, 4X, 5D
2070 PRINT USING 2080:B(4,1),V(4)
2080 IMAGE "      SHEAR HOR = ".S6D, 4X, 5D
2090 PRINT USING 2100:B(5,1),V(5)
2100 IMAGE "      SHEAR N-V = ".S6D, 4X, 5D
2110 PRINT USING 2120:B(6,1),V(6)
2120 IMAGE "      SHEAR E-V = ".S6D, 4X, 5D
2130 REM *** COMPUTE PRINCIPAL STRESSES AND ORIENTATIONS
2140 S(1,1)=B(1,1)
2150 S(2,1)=B(4,1)
2160 S(3,1)=B(6,1)
2170 S(1,2)=B(4,1)
2180 S(2,2)=B(2,1)
2190 S(3,2)=B(5,1)
2200 S(1,3)=B(6,1)
2210 S(2,3)=B(5,1)
2220 S(3,3)=B(3,1)
2230 REM *** COMPUTE INVARIANTS OF STRESS TENSOR
2240 I1=S(1,1)+S(2,2)+S(3,3)
2250 I2=S(1,1)*S(3,3)+S(2,2)*S(3,3)+S(1,1)*S(2,2)
2260 I4=S(2,3)**2+S(1,2)**2+S(1,3)**2
2270 I2=I2-I4
2280 MAT P= INV(S)
2290 I3=DET
2300 REM *** SOLVE CUBIC EQUATION BY TRIGONOMETRIC SUBSTITUTION
2310 Q1=-I1
2320 Q2=I2
2330 Q3=-I3
2340 Q4=(3*Q2-Q1*Q1)/3
2350 Q5=(2*Q1**3-9*Q1*Q2+27*Q3)/27

```

```

2360 Q6=-Q5/2/SQR(-Q4^3/27)
2370 Q7=ACS(Q6)/3
2380 X1=2*SQR(-Q4/3)*COS(Q7)
2390 X2=2*SQR(-Q4/3)*COS(Q7+120/R1)
2400 X3=2*SQR(-Q4/3)*COS(Q7+240/R1)
2410 Z(1,1)=X1-Q1/3
2420 Z(2,1)=X2-Q1/3
2430 Z(3,1)=X3-Q1/3
2440 REM *** END OF CUBIC SOLUTION
2450 REM *** SORT PRINCIPAL STRESSES
2460 Z1=MAX(Z(1,1),Z(2,1),Z(3,1))
2470 Z3=MIN(Z(1,1),Z(2,1),Z(3,1))
2480 Z(1,1)=Z1
2490 Z(2,1)=I1-Z1-Z3
2500 Z(3,1)=Z3
2510 REM *** END OF SORT
2520 REM *** COMPUTE DIRECTION COSINES OF PRINCIPAL STRESSES
2530 FOR N=1 TO 3
2540   MAT I= IDN
2550   MAT X= (Z(N,1))*I
2560   MAT Y= S-X
2570   N3=(Y(1,1)*Y(2,2)-Y(2,1) 2)/(Y(1,2)*Y(2,3)-Y(2,2)*Y(1,3))
2580   N2=-(Y(1,1)+Y(1,3)*N3)/Y(1,2)
2590   N4=SQR(1+N2*N2+N3*N3)
2600   P(1,N)=1/N4
2610   P(2,N)=N2/N4
2620   P(3,N)=N3/N4
2630   GOSUB 2970
2640   Z(N,2)=B5
2650   Z(N,3)=I5
2660 NEXT N
2670 REM *** RE-COMPUTE DIRECTION COSINES WITH RESPECT TO PRINCIPAL AXES
2680 REM *** TO COMPUTE STANDARD DEVIATIONS OF PRINCIPAL STRESSES
2690 FOR N=1 TO M
2700   L1=K(N,1)*P(1,1)+K(N,4)*P(2,1)+K(N,7)*P(3,1)
2710   M1=K(N,1)*P(1,2)+K(N,4)*P(2,2)+K(N,7)*P(3,2)
2720   N1=K(N,1)*P(1,3)+K(N,4)*P(2,3)+K(N,7)*P(3,3)
2730   L2=K(N,2)*P(1,1)+K(N,5)*P(2,1)+K(N,8)*P(3,1)
2740   M2=K(N,2)*P(1,2)+K(N,5)*P(2,2)+K(N,8)*P(3,2)
2750   N2=K(N,2)*P(1,3)+K(N,5)*P(2,3)+K(N,8)*P(3,3)
2760   L3=K(N,3)*P(1,1)+K(N,6)*P(2,1)+K(N,9)*P(3,1)
2770   M3=K(N,3)*P(1,2)+K(N,6)*P(2,2)+K(N,9)*P(3,2)
2780   N3=K(N,3)*P(1,3)+K(N,6)*P(2,3)+K(N,9)*P(3,3)
2790   J(N,1)=F(N,1)*L1*L1+F(N,2)*L2*L2+F(N,3)*L3*L3+F(N,4)*L1*L3
2800   J(N,2)=F(N,1)*M1*M1+F(N,2)*M2*M2+F(N,3)*M3*M3+F(N,4)*M1*M3
2810   J(N,3)=F(N,1)*N1*N1+F(N,2)*N2*N2+F(N,3)*N3*N3+F(N,4)*N1*N3
2820   J(N,4)=2*F(N,1)*L1*M1+2*F(N,2)*L2*M2+2*F(N,3)*L3*M3+F(N,4)*(L1*M3+L3*M1)
2830   J(N,5)=2*F(N,1)*M1*N1+2*F(N,2)*M2*N2+2*F(N,3)*M3*N3+F(N,4)*(M1*N3+M3*N1)
2840   J(N,6)=2*F(N,1)*N1*L1+2*F(N,2)*N2*L2+2*F(N,3)*N3*L3+F(N,4)*(N1*L3+N3*L1)
2850 NEXT N
2860 K9=2
2870 GOTO 1540
2880 PRINT
2890 PRINT " PRINCIPAL STRESSES"
2900 PRINT " STRESS STD. DEV. BEARING INCL."
2910 FOR N=1 TO 3
2920   PRINT USING 2940:B(N,1),V(N),Z(N,2),Z(N,3)
2930 NEXT N
2940 IMAGE 3X,SDDDDDD,4X,DDDDDD,5X,DDD.D,3X,SDD.D
2950 REM END

```

```

2960 GOTO 3210
2970 REM *** COMPUTE BEARINGS AND INCLINATIONS
2980 REM *** OF PRINCIPAL STESSES
2990 IS=90-ACS(P(3,N))*R1
3000 IF P(1,N)<>0 THEN 3090
3010 IF P(2,N)>0 THEN 3050
3020 IF P(2,N)<0 THEN 3070
3030 B5=999
3040 GOTO 3200
3050 B5=0
3060 GOTO 3200
3070 B5=180
3080 GOTO 3200
3090 B5=ATN(P(1,N)/P(2,N))*R1
3100 IF P(2,N)<0 THEN 3140
3110 IF P(1,N)>0 THEN 3150
3120 B5=360+B5
3130 GOTO 3150
3140 B5=180+B5
3150 IF IS<0 THEN 3180
3160 IS=-IS
3170 B5=180+B5
3180 IF B5<360 THEN 3200
3190 B5=B5-360
3200 RETURN
3210 END

```

```

10  REM ***          PROGRAM LISTING FOR CSIRO
20  REM ***
30  REM *** *****
40  REM ***
50  REM *** A PROGRAM FOR THE CALCULATION OF STRESS COMPONENTS, PRINCIPAL
60  REM *** STRESSES, AND THEIR STANDARD DEVIATIONS FROM STRAIN READINGS FROM
70  REM *** THE csiro hollow inclusion stress cell IN A SINGLE BOREHOLE.
80  REM ***
90  REM *** PROGRAM MODIFIED FROM CSIRO BY WILLIAM K. SMITH BY P. POWERS
100 REM *** JUNE 6, 1985.
110 REM *** PROGRAM WILL RUN WITH THE HP-9836 WITH BASIC 3.0, OR BASIC
120 REM *** 2.0 WITH THE AP2_1 EXTENSION.
130 REM ***
140 REM *** LITTLE OR NO MODIFICATION WAS REQUIRED TO RUN THE PROGRAM ON
150 REM *** THE HP-9836. ONE SIGNIFICANT MODIFICATION WAS THAT THE HP-9836
160 REM *** DOES NOT DIFFERENTIATE BETWEEN ARRAY NAMES AND SIMPLE VARIABLE
170 REM *** NAMES THAT ARE THE SAME, THEREBY PRODUCING AN ERROR. NEW SIMPLE
180 REM *** VARIABLES HAVE BEEN INTRODUCED INTO THE PROGRAM TO COMPENSATE
190 REM *** FOR THIS DIFFERENCE (I.E. VARIABLE 'I' WAS CHANGED TO 'I1').
200 REM ***
210 REM *** ***** REFERENCES *****
220 REM ***
230 REM *** PANEK, LOUIS A., 1966, Calculation of the average ground stress
240 REM *** components from measurements of the diametral deformation of
250 REM *** a drill hole: U.S. Bureau of Mines Rept. of Inv. 6732, 41 p.;
260 REM *** or ASTM S.T.P. 402, p. 106-132.
270 REM *** WORDTNIKI, G., and WALTON, R., 1976, Triaxial "hollow inclusion"
280 REM *** gauges for determination of rock stresses in situ: Common-
290 REM *** wealth Sci. and Ind. Res. Org. (AUSTRALIA) Div., Applied
300 REM *** Geomechanics Research Paper 275; or Proc. Symp. on Investiga-
310 REM *** tion of stress in rock - Advances in stress measurement,
320 REM *** Sydney, 1976, p. 1-8 (pre-print).
330 REM *** FIELD Manual for CSIRO hollow inclusion gauge (includes FORTRAN
340 REM *** data reduction program).
350 REM *** SMITH, WILLIAM K., 1982, Two BASIC computer programs for the
360 REM *** determination of in-situ stresses using the CSIRO hollow
370 REM *** inclusion gage and the USBM borehole deformation gage:
380 REM *** U.S. Geological Survey Open-File Rept. 82-489, 40 p.
390 REM ***
400 REM *** *****
410 REM ***
420 REM *** COORDINATE SYSTEMS:
430 REM *** X-AXIS IS POSITIVE EAST
440 REM *** Y-AXIS IS POSITIVE NORTH
450 REM *** Z-AXIS IS POSITIVE UP
460 REM ***
470 REM *** H1-AXIS IS HORIZONTAL TO THE RIGHT WHEN LOOKING INTO THE HOLE
480 REM *** H2-AXIS COINCIDES WITH THE BOREHOLE AXIS.
490 REM *** H3-AXIS IS IN A VERTICAL PLAN, POINTS GENERALLY UPWARD
500 REM ***
510 REM *** BEARINGS ARE IN DEGREES CLOCKWISE FROM THE NORTH (AZIMUTH).
520 REM *** INCLINATIONS ARE IN DEGREES FROM HORIZONTAL, POSITIVE UPWARD.
530 REM *** BEARING OF A VERTICAL HOLE IS 90 DEGREES COUNTER-CLOCKWISE
540 REM *** FROM THE AZIMUTH OF THE H1-AXIS, AS DETERMINED FROM THE
550 REM *** ORIENTATION OF THE SETTING TOOL.
560 REM *** UNITS FOR THE STRESS VALUES IN THE OUTPUT ARE SAME AS THOSE FOR
570 REM *** USED FOR YOUNG'S MODULUS IN THE INPUT.
580 REM *** ORIENTATION ANGLES ALPHA & BETA AND CORRECTION FACTORS K1-K4 ARE
590 REM *** INITIALIZED IN DATA STATEMENTS.

```

```

600 REM *** ALPHA AND BETA ANGLES WILL NOT CHANGE AS LONG AS STANDARD
610 REM *** INSTALLATION PROCEDURES ARE USED. CORRECTION FACTORS MAY CHANGE
620 REM *** DEPENDING ON THE ROCK/EPOXY MODULUS RATIO.
630 REM ***
640 REM *** *****
650 OPTION BASE 1
660 PRINTER IS 1
670 DIM A(6,6),B(6,1),C(6,6),D(9,2),E(9),F(9,6),G(6,1),I(3,3),J(9,6),K(9,9)
680 DIM P(3,3),S(3,3),T(9,2),U(9),V(6),X(3,3),Y(3,3),Z(3,3),H(9)
690 R1=57.29578
700 K9=1
710 PRINT "INPUT NUMBER OF STRAIN READINGS TO BE AVERAGED"
720 INPUT M
730 PRINT CHR$(12)
740 PRINT "INPUT ELASTIC MODULUS AND POISSON'S RATIO"
750 INPUT E1,P1
760 PRINT CHR$(12)
770 PRINT "INPUT BEARING AND INCLINATION OF BOREHOLE"
780 INPUT B2,I2
790 B1=B2/R1
800 I1=I2/R1
810 REM *** READ ORIENTATION ANGLES ALPHA AND BETA
820 REM *** AND CORRECTION FACTORS K1 THROUGH K4
830 FOR N=1 TO 9
840 READ D(N,1)
850 NEXT N
860 FOR N=1 TO 9
870 READ D(N,2)
880 NEXT N
890 READ K1,K2,K3,K4
900 DATA 322.9,300,300,163.6,163.6,180,82.9,60,60
910 DATA 0.90,45,45,135,90,0,90,45
920 DATA 1,12,1,13,1,08,0,91
930 REM *** CONVERT ALPHA AND BETA TO RADIANS
940 MAT T= (1/R1)*D
950 PRINT CHR$(12)
960 PRINT "INPUT ";M;"GAGE NUMBERS AND CORRESPONDING MICROSTRAINS"
970 FOR N=1 TO M
980 INPUT H(N),E(N)
990 U(N)=E(N)*1.0E-6
1000 NEXT N
1010 PRINT CHR$(12)
1020 PRINTER IS 701
1030 PRINT "*** CSIRO HOLLOW INCLUSION STRESS CELL DATA REDUCTION ***"
1040 PRINT
1050 PRINT " MICRO-"
1060 PRINT " GAGE NO. STRAIN ALPHA BETA"
1070 FOR N=1 TO M
1080 PRINT USING 1090:H(N),E(N),D(H(N),1),D(H(N),2)
1090 IMAGE (4X,D,6X,S6D,5X,3D,D,4X,3D,D)
1100 NEXT N
1110 PRINT
1120 PRINT " BOREHOLE BEARING = ";B2;" INCLINATION = ";I2
1130 PRINT " MODULUS = ";E1;" POISSON'S RATIO = ";P1
1140 PRINT
1150 L1=COS(B1)
1160 M1=-SIN(B1)
1170 N1=0
1180 L2=SIN(B1)*COS(I1)
1190 M2=COS(B1)*COS(I1)

```

```

1200 N2=SIN(I1)
1210 L3=-SIN(B1)*SIN(I1)
1220 M3=-COS(B1)*SIN(I1)
1230 N3=COS(I1)
1240 FOR N=1 TO M
1250   K(N,1)=L1
1260   K(N,2)=L2
1270   K(N,3)=L3
1280   K(N,4)=M1
1290   K(N,5)=M2
1300   K(N,6)=M3
1310   K(N,7)=N1
1320   K(N,8)=N2
1330   K(N,9)=N3
1340   F1=COS(2*T(H(N),1))
1350   F2=COS(2*T(H(N),2))
1360   F3=SIN(2*T(H(N),1))
1370   F4=SIN(2*T(H(N),2))
1380   F5=P1*P1
1390   F(N,1)=((1-F5)*F1*(1-F2)*K2+.5*(K1-P1)-.5*(K1+P1)*F2)/E1
1400   F(N,2)=(.5*(1-K4*P1)+.5*(1+K4*P1)*F2)/E1
1410   F(N,3)=((F5-1)*F1*(1-F2)*K2+.5*(K1-P1)-.5*(K1+P1)*F2)/F1
1420   F(N,4)=2*(1+P1)*COS(T(H(N),1))*F4*K3/E1
1430   F(N,5)=-2*(1+P1)*SIN(T(H(N),1))*F4*K3/E1
1440   F(N,6)=2*(F5-1)*F3*(1-F2)*K2/E1
1450 NEXT N
1460 FOR N=1 TO 9
1470   J(N,1)=F(N,1)*L1*L1+F(N,2)*L2*L2+F(N,3)*L3*L3+F(N,4)*L1*L2+F(N,5)*L2*L3+
F(N,6)*L1*L3
1480   J(N,2)=F(N,1)*M1*M1+F(N,2)*M2*M2+F(N,3)*M3*M3+F(N,4)*M1*M2+F(N,5)*M2*M3+
F(N,6)*M1*M3
1490   J(N,3)=F(N,1)*N1*N1+F(N,2)*N2*N2+F(N,3)*N3*N3+F(N,4)*N1*N2+F(N,5)*N2*N3+
F(N,6)*N1*N3
1500   J(N,4)=2*F(N,1)*L1*M1+2*F(N,2)*L2*M2+2*F(N,3)*L3*M3+F(N,4)*(L1*M2+L2*M1)
+F(N,5)*(L2*M3+L3*M2)+F(N,6)*(L1*M3+L3*M1)
1510   J(N,5)=2*F(N,1)*M1*N1+2*F(N,2)*M2*N2+2*F(N,3)*M3*N3+F(N,4)*(M1*N2+M2*N1)
+F(N,5)*(M2*N3+M3*N2)+F(N,6)*(M1*N3+M3*N1)
1520   J(N,6)=2*F(N,1)*L1*N1+2*F(N,2)*L2*N2+2*F(N,3)*L3*N3+F(N,4)*(L1*N2+L2*N1)
+F(N,5)*(L2*N3+L3*N2)+F(N,6)*(L1*N3+L3*N1)
1530 NEXT N
1540 REM *** CLEAR A(I,J) AND G(J) ARRAYS
1550 FOR J1=1 TO 6
1560   G(J1,1)=0
1570   FOR I1=1 TO 6
1580     A(I1,J1)=0
1590   NEXT I1
1600 NEXT J1
1610 REM *** COMPUTE [A] AND [G] ARRAYS
1620 FOR K1=1 TO M
1630   FOR J1=1 TO 6
1640     G(J1,1)=G(J1,1)+U(K1)*J(K1,J1)
1650     FOR I1=1 TO 6
1660       A(I1,J1)=A(I1,J1)+J(K1,I1)*J(K1,J1)
1670     NEXT I1
1680   NEXT J1
1690 NEXT K1
1700 MAT C= INV(A)
1710 MAT B= C*G
1720 REM *** COMPUTE STATISTICAL PARAMETERS
1730 IF K9=2 THEN 1950

```

```

1740 S1=0
1750 FOR K1=1 TO M
1760 S1=S1+U(K1)*U(K1)
1770 NEXT K1
1780 S2=B(1,1)*G(1,1)+B(2,1)*G(2,1)+B(3,1)*G(3,1)+B(4,1)*G(4,1)+B(5,1)*G(5,1)+B
(6,1)*G(6,1)
1790 S3=S1-S2
1800 R2=1-S3/S1
1810 R=SQR(R2)
1820 S4=SQR(S3/(M-6))
1830 PRINT USING 1890:S1
1840 PRINT USING 1900:S2
1850 PRINT USING 1910:S3
1860 PRINT USING 1920:S4
1870 PRINT USING 1930:R2
1880 PRINT USING 1940:R
1890 IMAGE " STRAIN SUM OF SQUARES = ".D.4DE
1900 IMAGE " [B],[G] = ".D.4DE
1910 IMAGE " RESIDUAL SUM OF SQUARES = ".D.4DE
1920 IMAGE " STD. DEV OF FITTED DATA = ".D.4DE
1930 IMAGE " GOODNESS OF FIT (R**2) = ".D.5D
1940 IMAGE " CORRELATION COEFFICIENT = ".D.5D
1950 FOR I1=1 TO 6
1960 V(I1)=S4*SQR(C(I1,I1))
1970 NEXT I1
1980 IF K9=2 THEN 2820
1990 PRINT
2000 PRINT " STRESS COMPONENTS STD. DEV."
2010 PRINT USING 2020:B(2,1),V(2)
2020 IMAGE " NORMAL N-S = ",S3D.2D,4X,3D.2D
2030 PRINT USING 2040:B(1,1),V(1)
2040 IMAGE " NORMAL E-W = ",S3D.2D,4X,3D.2D
2050 PRINT USING 2060:B(3,1),V(3)
2060 IMAGE " NORMAL VER = ",S3D.2D,4X,3D.2D
2070 PRINT USING 2080:B(4,1),V(4)
2080 IMAGE " SHEAR HORZ = ",S3D.2D,4X,3D.2D
2090 PRINT USING 2100:B(5,1),V(5)
2100 IMAGE " SHEAR N-V = ",S3D.2D,4X,3D.2D
2110 PRINT USING 2120:B(6,1),V(6)
2120 IMAGE " SHEAR E-V = ",S3D.2D,4X,3D.2D
2130 REM *** COMPUTE PRINCIPAL STRESSES AND ORIENTATIONS
2140 S(1,1)=B(1,1)
2150 S(2,1)=B(4,1)
2160 S(3,1)=B(6,1)
2170 S(1,2)=B(4,1)
2180 S(2,2)=B(2,1)
2190 S(3,2)=B(5,1)
2200 S(1,3)=B(6,1)
2210 S(2,3)=B(5,1)
2220 S(3,3)=B(3,1)
2230 REM *** COMPUTE INVARIANTS OF STRESS TENSOR
2240 I1=S(1,1)+S(2,2)+S(3,3)
2250 I2=S(1,1)*S(3,3)+S(2,2)*S(3,3)+S(1,1)*S(2,2)
2260 I4=S(2,3)^2+S(1,2)^2+S(1,3)^2
2270 I2=I2-I4
2280 MAT P= INV(S)
2290 I3=DET
2300 REM *** SOLVE CUBIC EQUATION BY TRIGONOMETRIC SUBSTITUTION
2310 Q1=-I1
2320 Q2=I2

```

```

2330 Q3=-I3
2340 Q4=(3*Q2-Q1*Q1)/3
2350 Q5=(2*Q13-9*Q1*Q2+27*Q3)/27
2360 Q6=-Q5/2/SQR(-Q43/27)
2370 Q7=ACS(Q6)/3
2380 X1=2*SQR(-Q4/3)*COS(Q7)
2390 X2=2*SQR(-Q4/3)*COS(Q7+120/R1)
2400 X3=2*SQR(-Q4/3)*COS(Q7+240/R1)
2410 Z(1,1)=X1-Q1/3
2420 Z(2,1)=X2-Q1/3
2430 Z(3,1)=X3-Q1/3
2440 REM *** END OF CUBIC SOLUTION
2450 REM *** SORT PRINCIPAL STRESSES
2460 Z1=MAX(Z(1,1),Z(2,1),Z(3,1))
2470 Z3=MIN(Z(1,1),Z(2,1),Z(3,1))
2480 Z(1,1)=Z1
2490 Z(2,1)=I1-Z1-Z3
2500 Z(3,1)=Z3
2510 REM *** END OF SORT
2520 REM *** COMPUTE DIRECTION COSINES OF PRINCIPAL STRESSES
2530 FOR N=1 TO 3
2540   MAT I= IDN
2550   MAT X= (Z(N,1))*I
2560   MAT Y= S-X
2570   N3=(Y(1,1)*Y(2,2)-Y(2,1)2)/(Y(1,2)*Y(2,3)-Y(2,2)*Y(1,3))
2580   N2=-(Y(1,1)+Y(1,3)*N3)/Y(1,2)
2590   N4=SQR(1+N2*N2+N3*N3)
2600   P(1,N)=1/N4
2610   P(2,N)=N2/N4
2620   P(3,N)=N3/N4
2630   GOSUB 2910
2640   Z(N,2)=B5
2650   Z(N,3)=I5
2660 NEXT N
2670 REM *** RE-COMPUTE DIRECTION COSINES WITH RESPECT TO PRINCIPAL AXES
2680 REM *** TO COMPUTE STANDARD DEVIATIONS OF PRINCIPAL STRESSES
2690 FOR N=1 TO M
2700   L1=K(N,1)*P(1,1)+K(N,4)*P(2,1)+K(N,7)*P(3,1)
2710   M1=K(N,1)*P(1,2)+K(N,4)*P(2,2)+K(N,7)*P(3,2)
2720   N1=K(N,1)*P(1,3)+K(N,4)*P(2,3)+K(N,7)*P(3,3)
2730   L2=K(N,2)*P(1,1)+K(N,5)*P(2,1)+K(N,8)*P(3,1)
2740   M2=K(N,2)*P(1,2)+K(N,5)*P(2,2)+K(N,8)*P(3,2)
2750   N2=K(N,2)*P(1,3)+K(N,5)*P(2,3)+K(N,8)*P(3,3)
2760   L3=K(N,3)*P(1,1)+K(N,6)*P(2,1)+K(N,9)*P(3,1)
2770   M3=K(N,3)*P(1,2)+K(N,6)*P(2,2)+K(N,9)*P(3,2)
2780   N3=K(N,3)*P(1,3)+K(N,6)*P(2,3)+K(N,9)*P(3,3)
2790 NEXT N
2800 K9=2
2810 GOTO 1460
2820 PRINT
2830 PRINT " PRINCIPAL STRESSES"
2840 PRINT "      STRESS STD. DEV. BEARING INCL."
2850 FOR N=1 TO 3
2860   PRINT USING 2880:B(N,1),V(N),Z(N,2),Z(N,3)
2870 NEXT N
2880 IMAGE 3X,S3D,2D,4X,2D,2D,5X,3D,D,3X,S2D,D
2890 REM THE END
2900 GOTO 3150
2910 REM *** COMPUTE BEARINGS AND INCLINATIONS
2920 REM *** OF PRINCIPAL STRESSES

```



```

2930 IS=90-ACS(P(3,N))*R1
2940 IF P(1,N)<>0 THEN 3030
2950 IF P(2,N)>0 THEN 2990
2960 IF P(2,N)<0 THEN 3010
2970 B5=999
2980 GOTO 3140
2990 B5=0
3000 GOTO 3140
3010 B5=180
3020 GOTO 3140
3030 B5=ATN(P(1,N)/P(2,N))*R1
3040 IF P(2,N)<0 THEN 3080
3050 IF P(1,N)>0 THEN 3090
3060 B5=360+B5
3070 GOTO 3090
3080 B5=180+B5
3090 IF IS<0 THEN 3120
3100 IS=-IS
3110 B5=180+B5
3120 IF B5<360 THEN 3140
3130 B5=B5-360
3140 RETURN
3150 END

```

## REFERENCES

- Fitzpatrick, J., 1962, Biaxial device for determining the modulus of elasticity in stress-relief cores: U.S. Bureau of Mines Report of Investigations 7015, 38 p.
- Merrill, R. H., 1967, Three-component borehole deformation gage for determining the stress in rock: U.S. Bureau of Mines Report of Investigations 7015, 38 p.
- Panek, L. A., 1966, Calculation of the average ground stress components from measurements of the diametral deformation of a drill hole: American Society for Testing and Materials Special Technical Publication 402, p. 106-132.
- Smith, W. K., 1982, Two BASIC computer programs for the determination of in situ stresses using the CSIRO hollow inclusion stress cell and the USBM borehole deformation gage [Part II of In situ geomechanics of crystalline and sedimentary rocks]: U.S. Geological Survey Open-File Report 82-489, 40 p.
- Worotnicki, G., and Walton, R. J., 1976, Triaxial "hollow-inclusion" gauges for the determination of rock stresses in situ, in Investigation of stress in rock--Advances in stress measurement, International Society of Rock Mechanics Symposium, Sidney, Australia, National Conference Publication 76/4, Supplement, p. 1-8.